



A PUFsecurity Alliance

# Digital Logic Design

- Lecture 1
- Introduction ■


2025 Spring



# Agenda ■

1. Introduction
2. RTL basic
3. Testbench basic

# From Code to Chip .



# Hardware Description Language ■


- Hardware Description Language (HDL)
  - Describe the structure and behavior of electronic circuits
  - Writing HDL is just like drawing a circuit.
- Important Note
  - HDL is **not** a software code

Software language	HDL
Describe program, <u>Execute in order</u>	Describe hardware, <u>Execute in parallel</u>
Can be run on specific platforms like CPU or MCU	Can be general purpose IC (CPU, MCU) or application specific IC (ASIC)
Software code Binary code Machine code	Register-Transfer Level Gate level Transistor level

# Hardware Description Language ■


## ■ Common HDL

- Transistor Level
  - HSPICE
  - PSPICE



Static CMOS logic inverter

[Inverter \(logic gate\) - Wikipedia](#)




A CMOS IC implementation of a dynamic edge-triggered flip-flop with reset

[Flip-flop \(electronics\) - Wikipedia](#)


- Gate Level
- Register-Transfer Level (RTL)
  - VHDL
  - Verilog



This Lecture



Traditional NOT gate (inverter) symbol




D flip-flop symbol

# History of Verilog

- First created in 1983
- IEEE 1364-1995: Verilog-95 (.v)
  - Fundamental syntax
- IEEE 1364-2001: Verilog-2001
  - Improve readability and functionality
- IEEE 1364-2005: Verilog-2005
  - Minor improvements
- IEEE 1800-2009: SystemVerilog (.sv)
  - More strict and abstract way for describing hardware
  - Powerful testbench syntax

This Lecture



<https://nguyenquanicd.blogspot.com/2017/08/verilog-nao-la-verilog-hoac-system.html>

# Agenda ■

1. Introduction
2. RTL basic
3. Testbench basic



# Verilog Execution Order ■

- In Verilog, all code blocks are executed in parallel
- design
  - All hardware (assignment or procedural blocks) were executed in parallel

```
6 assign a = e;  
7  
8 always@(*)begin  
9     b = |f;  
10 end
```

```
12 assign c = g;  
13  
14 always@(*)begin  
15     d = &h;  
16 end
```

- testbench
  - All procedural blocks of test code were start simultaneously

```
19 initial begin  
20     a = 1'b0;  
21 end
```

```
23 initial begin  
24     b = 1'b1;  
25 end
```

# Verilog Execution Order Inside “begin...end”

## ■ Blocking assignment “=”

- Like software code
- Executed in order

```
69  ////////////////////////////////////
70  //blocking assignment
71  integer ba;
72  integer bb;
73
74  initial begin
75      #1;
76      //init value
77      ba = 1;
78      bb = 2;
79
80      #10;//delay 10ns
81      ba = bb;
82      bb = ba;
83
84      $display("-----");
85      $display("ba:%d",ba); //2
86      $display("bb:%d",bb); //2
87      $display("-----");
88  end
```

## ■ Non-blocking assignment “<=”

- Describe the circuit behavior
- Executed in parallel



```
92  ////////////////////////////////////
93  //non-blocking assignment
94  integer na;
95  integer nb;
96
97  initial begin
98      @(negedge rst_n);
99      //init value
100     na <= 1;
101     nb <= 2;
102
103     @(posedge clk);
104     na <= nb;
105     nb <= na;
106     #1;//print value a little later
107     $display("-----");
108     $display("na:%d",na); //2
109     $display("nb:%d",nb); //1
110     $display("-----");
111  end
```

# Basic Components of a Digital Circuit ■



- Combinational circuits
  - Logic gates
  - The output depends directly on the input, and be independent on time event signal
- Sequential circuits
  - The output depends not only on the current input, but also on time event signal
  - It has ability to store previous state.

# Sequential circuits ■

- Latch, level-triggered
  - Output will only change when the enable signal occurs




- Flip-Flop (also called register), edge-triggered
  - Output will only change when the edge of clock/reset occurs



# Signal Declaration vs Design Intent



## ■ Combinational circuits

- “wire” type for continuous assignment
- “reg” type for procedural block



## ■ Sequential circuit

- “reg” type for procedural block
- Flipflop
- “reg” type for procedural block
- Latch



```
1 wire c0;//cmb_0
2 assign c0 = (sel)? x: y;
3
4 reg c1;//cmb_1
5 always@*begin
6     if(sel) c1 = x;
7     else    c1 = y;//complete statement
8 end                                     //becomes combinational
9
10
11
12
13 reg f0;//flipflop_0
14 always@(posedge clk or negedge rst_n)begin
15     if(!rst_n) f0 <= 1'b0;
16     else if(..) f0 <= ...;
17 end
18
19
20
21
22 reg l0;//latch_0
23 always@*begin
24     if(en) l0 = x;//incomplete statement
25 end                                     //becomes latch
```

- Describe the logical behavior of the circuit,  
in fact, there will be a time delay in each circuit after synthesis to different technology node.

# 4-state Variable in Verilog ■

- To describe the circuit behavior, the variable (each bits) has 4 states in Verilog
  - “0” : logical 0 (Physically connected to VSS/GND)
  - “1” : logical 1 (Physically connected to VDD)
  - “z” : high-impedance (Physically not connected or tri-state)
  - “x” : unknown (Simulator cannot determine the value)
- Value after declaration
  - wire : “z”
  - reg : “x”
- Value propagation
  - wire : become “x” if RHS has “x” or “z”, and the simulator cannot determine the value.
  - reg : become “x” if RHS has “x” or “z”, and the simulator cannot determine the value.

(RHS: Right Hand Side)



Avoid “x” in  
design after  
reset signal

# Coding Style ■

- Verilog syntax is very loose and can write a variety of circuit behavior, but the circuit of such behavior may not exist.
- Bad **coding styles** can lead to inconsistent results in different EDA tools.
  - Simulator vs Simulator (ex: NC-verilog, VCS)
    - In design or testbench
    - Weird waveform (blocking, non-blocking racing issue)
  - Simulator vs Synthesizer (ex: NC-verilog, Design Compiler)
    - In design
    - Unexpected circuit (latch or other strange circuit)



```
1 reg ff1;  
2 reg ff2;  
3 always@(posedge clk or negedge clk)begin  
4     ff1 = ff2;  
5     ff2<= ff1;  
6 end
```

# Avoid Latches in Design ■

- Problem of latch
  - level-triggered components are sensitive to noise interference.
  - Difficult to analyze timing in the synthesis stage.
  - Only some special purposes will intentionally use latch.

- Important Note
  - Most of the time, avoid latches in design stage



```
4 reg c1; //cmb_1
5 always@*begin
6     if(sel) c1 = x;
7     else    c1 = y; //complete statement
8 end                //becomes combinational
```



```
22 reg l0; //latch_0
23 always@*begin
24     if(en) l0 = x; //incomplete statement
25 end                //becomes latch
```




# Coding Style for Combinational Circuit ■

- Design intent: 3-to-1 mux



```
3 //bad coding style,
4 //incomplete statement
5 //y will hold it's value when sel=2'b11, it becomes latch
6 always@*begin
7     case(sel)
8         2'b00:    y=a;
9         2'b01:    y=b;
10        2'b10:    y=c;
11    endcase
12 end
```



```
15 //bad coding style,
16 //incomplete statement, with EDA tools specific comment
17 //y is 3-to-1 mux only when using synopsys tool
18 always@*begin
19     case(sel)//synopsys full_case
20         2'b00:    y=a;
21         2'b01:    y=b;
22         2'b10:    y=c;
23     endcase
24 end
```



```
27 //correct coding style
28 //use default (or else) statement
29 //y is 3-to-1 mux in all situations
30 always@*begin
31     case(sel)
32         2'b00:    y=a;
33         2'b01:    y=b;
34         default: y=c;
35     endcase
36 end
```

# Coding Style for Combinational Circuit ■

## ■ Design intent: 3-to-1 mux

### – Verilog-1995




```
62 //Verilog-1995
63 //Sensitive lists should be specifically filled out
64 reg y;
65 always@(a or b or c or sel)begin
66     if(sel==2'b00)    y=a;
67     else if(sel==2'b01) y=b;
68     else              y=c;
69 end
```

### – Verilog-2001



```
70 //Verilog-2001
71 //Sensitive lists can be *
72 reg y;
73 always@*begin
74     if(sel==2'b00)    y=a;
75     else if(sel==2'b01) y=b;
76     else              y=c;
77 end
```



# Coding Style for Sequential Circuit ■

## ■ Design intent: D-flipflop with selection

- What will happen if `rst_n == 0` and `en == 1`?




```
3 //bad coding style,
4 //control a variable in different blocks
5 reg q;
6 always@(posedge clk or negedge rst_n)begin
7     if(!rst_n) q<=1'b0;
8 end
9 always@(posedge clk or negedge rst_n)begin
10    if(en)      q<=d;
11 end
```

- Describe the signal behavior in one always block



```
14 //correct coding style
15 //control a variable in only-1 block
16 reg q;
17 always@(posedge clk or negedge rst_n)begin
18     if(!rst_n) q<=1'b0;
19     else if(en)q<=d;
20 end
```




# Module ■

- “module” represents a group of related circuits in Verilog
- Can be either combinational or sequential circuits or both



**module** module\_name

- Parameter list
  - Configurable when using this module
  - Constant at compile time
- Port list
  - input (Usually have)
  - output (Usually have)
  - inout (special circuits)
- Internal signal declaration
- Circuit implementation

**endmodule**



# Module




```

1 module addr#( //name of this module
2     //parameter declaration
3     parameter WIDTH = 8
4 )
5     //port declaration
6     input    [WIDTH-1:0] a    ,//input a
7     input    [WIDTH-1:0] b    ,//input b
8     output reg [WIDTH :0] s    ,//seq output of (a + b)
9     input    clk               ,
10    input    rst_n              ,
11 );
12
13 //internal signal declaration
14 wire [WIDTH :0] ns;//cmb value of (a+b)
15
16 //circuit implementation
17 //cmb
18 assign ns = a + b;
19 //seq
20 always@(posedge clk or negedge rst_n)begin
21     if(!rst_n) s <= {WIDTH+1{1'b0}};
22     else      s <= ns;
23 end
24
25 endmodule //end of this module
    
```


# Module Instance and Connection

```
1 module addr4#(  
2     //parameter declaration  
3     parameter IN_WIDTH = 8  
4 )(  
5     //port declaration  
6     input  [IN_WIDTH-1:0] in1  ,  
7     input  [IN_WIDTH-1:0] in2  ,  
8     input  [IN_WIDTH-1:0] in3  ,  
9     input  [IN_WIDTH-1:0] in4  ,  
10    output [IN_WIDTH+1:0] out  ,  
11    input  clk  ,  
12    input  rst_n  
13 );  
14  
15 //internal signal declaration  
16 wire [IN_WIDTH :0] s1;  
17 wire [IN_WIDTH :0] s2;  
18  
19 //sub-module instancec  
20 addr#(.WIDTH(IN_WIDTH ))I_ADDR_1(.a(in1), .b(in2), .s(s1 ), .clk(clk), .rst_n(rst_n));  
21 addr#(.WIDTH(IN_WIDTH ))I_ADDR_2(.a(in3), .b(in4), .s(s2 ), .clk(clk), .rst_n(rst_n));  
22 addr#(.WIDTH(IN_WIDTH+1))I_ADDR_3(.a(s1 ), .b(s2 ), .s(out), .clk(clk), .rst_n(rst_n));  
23  
24 endmodule
```



# Signal Declaration, wire? reg? ■

- Module
  - Internal signal
    - Flipflop – **reg**
    - Combinational – **reg** or **wire**
  - IO
    - Input – **wire** (leave it blank)
    - Output – **reg** or **wire** (leave it blank)
- Sub-module instance
  - Connect to input – **wire** or **reg**
  - Connect to output – **wire**



What does  
the circuit  
look like?

- Having the circuit in your mind,  
It is obvious what type to declare.

# Coding Style for Port Declaration ■

## ■ Module port declaration

- Verilog-1995
- Many repetitive declarations
- Verilog-2001
- Simplified declarations
- add local parameter



```
1 //Verilog-1995 port declaration
2 module addr(a, b, s, clk, rst_n);
3     parameter WIDTH = 8;
4
5     input      [WIDTH-1:0] a    ;
6     input      [WIDTH-1:0] b    ;
7     output     [WIDTH  :0] s    ;
8     input      clk              ;
9     input      rst_n            ;
10
11     reg        [WIDTH  :0] s    ;
12
13     //...
14     //...
15
16 endmodule
```



```
19 //Verilog-2001 port declaration
20 module addr#(
21     parameter WIDTH = 8
22 )(
23     input      [WIDTH-1:0] a    ,
24     input      [WIDTH-1:0] b    ,
25     output reg  [WIDTH  :0] s    ,
26     input      clk              ,
27     input      rst_n            ,
28 );
29
30     localparam PIPENUM = 1;
31     //...
32     //...
33
34 endmodule
```



# Coding Style for Port Connection ■

## ■ Sub-module instance port connection

- by order (implicit)



```
40 //Verilog-1995 port connection, by order
41 addr#(IN_WIDTH) I_ADDR_1(in1, in2, s1, clk, rst_n);
```



- by name (explicit)



```
43 //Verilog-2001 port connection, by name
44 addr#(.WIDTH(IN_WIDTH))I_ADDR_1(.a(in1), .b(in2), .s(s1 ), .clk(clk), .rst_n(rst_n));
```


# Module IO Behavior

- When is the correct “out” value available?
- In order to easy integration of different modules, the IO waveform must be specified.



# Common Types of IO Protocol .


- Add control signals next to data signal.
  - The transmitter (tx) module
    - sends a control signal with data (data is optional)
  - The receiver (rx) module
    - replies a control signal with response (response is optional).
- Control signal types
  - Level (high / low)
  - Pulse (high within a clock period)
- Common types of IO protocol
  - Pulse, Pulse
  - Level, Pulse
  - Level, Level
  - Pulse, Level



# Common Types of IO Protocol ■

## ■ Request (Pulse), Acknowledgement (Pulse)


- TX sends a request (req) with optional data (d)
- RX replies an acknowledgement (ack) with an optional response signal (r)



# Common Types of IO Protocol ■

## ■ Request (Level), Acknowledgement (Pulse)


- TX sends a request (req) with optional data (d)
- RX replies an acknowledgement (ack) with an optional response signal (r)



# Common Types of IO Protocol ■

## ■ Valid (Level), Ready (Level)


- TX pulls up a valid signal (vld) to indicate that the output data (d) is valid.
- RX pulls up a ready signal (rdy) to indicate that it is ready to receive data (d).
- When both valid and ready are high it means the data has been successfully transferred.
- This protocol is often used to send continuous data
- The valid and ready signals must be independent from each other to avoid deadlock.



# Common Types of IO Protocol ■

## ■ Start (pulse), Busy (Level)

- TX sends a start signal (start\_p) with optional data (d)
- RX pulls up the busy signal (busy) and starts the internal operation when the operation is complete puts down the busy along with the response (r).
- This protocol is usually used for the controller module to operate the calculation module.



# Agenda ■

1. Introduction
2. RTL basic
3. Testbench basic




# Basic Components of Testbench

## ■ Testbench (tb\_top.sv)

```
module tb_top;
```

- Dump waveform
- DUT (Design Under Test) instance
- Generate clock, reset signal
- Drive DUT input (test pattern)
- Monitor DUT output
- Behavioral model of DUT
- Checker to compare answer
- ...

```
endmodule
```



# Basic Syntax of Testbench

- Procedural block (initial)
  - System task (\$)
    - \$random
    - \$display
    - \$finish
  - Time related
    - delay a specific time (#)
    - wait for event (@)
    - wait for condition (wait)
  - Branch
    - if-else
    - switch-case
  - Loop
    - for, while, repeat, forever

- begin ... end
  - A block of code in a branch or a loop



```
9      initial begin
10          a = 0;
11          if(cond)begin
12              a = a +1;//in if-block
13              a = a +2;//in if-block
14          end
15          $display("a: %d",a);//a is 0 when cond==false
16      end
```



```
18     initial begin
19         a = 0;
20         if(cond)
21             a = a +1;//only 1 line is in if-block
22             a = a +2;//out of if-block
23
24         $display("a: %d",a);//a is 2 when cond==false
25     end
```

# Dump Waveform ■

```
4  //////////////////////////////////////
5  // DUMPWAVE
6  //////////////////////////////////////
7  `ifdef USE_FSDB
8  initial begin
9      if($test$plusargs("FSDB")) begin//+FSDB to enable
10         $fsdbDumpfile("wave.fsdb");
11         // $fsdbDumpvars(depth, instance);
12         $fsdbDumpvars();
13     end
14 end
15 `endif
```

- ``define` is a preprocessor directive  
Can enable or disable some part of code,  
Not recommended for use in design.
- Because `$fsdbDumpXXX` is a Synopsys specific command,  
if you use other simulator, it will cause compile error, so use ``ifdef ... `endif` to separate it,  
use `+define+USE_FSDB` to open this code when simulation.
- Also `+FSDB` to trigger `$test$plusargs("FSDB")` to enter `if statement` in line 9.

# Timeout Control ■

- In the simulation process, there may be some reasons causing the simulation cannot be finished, so, it is better to set a timeout value in each testbench.
- In this example, the default timeout is 1000ns, if you need a longer timeout, you can use **+T0=XXXX** to overwrite.

- The **#** sign means delay for some unit length of time.



```
//          unit/precision  
'timescale 1ns/1ps
```

```
17  //////////////////////////////////////  
18  // TIMEOUT  
19  //////////////////////////////////////  
20  integer timeout;  
21  initial begin  
22      if(!$value$plusargs("T0=%d",timeout)) begin//+T0=XXXX to overwrite  
23          timeout = 1000;//default  
24      end  
25      #timeout;  
26      $display("*****");  
27      $display("simulation timeout");  
28      $display("*****");  
29      $finish;  
30  end
```

# Generate Clock and Reset

- In a large chip, the clock and reset signals are usually generated internally, but in the case of a small module simulation, the clock and reset signals are generated by the testbench.

```
32  //////////////////////////////////////
33  // CLK RESET
34  //////////////////////////////////////
35  reg clk ;
36  reg rst_n;
37
38  initial begin
39      clk = 1'b1;
40      forever begin
41          #(5.0/2.0) clk = ~clk;
42      end
43  end
44  initial begin
45      rst_n = 1'b1;
46      #7;
47      rst_n = 1'b0;
48      #7;
49      rst_n = 1'b1;
50  end
```



# Instance of DUT .

- There must be a module instance to be tested in the testbench, called **DUT** (Design Under Test)
- Set the **parameter** and connect the io when instance the module.
- Usually, the input signal of DUT will be operated in the initial block of testbench, so, it will be declared as **reg** type
- Testbench will check the output of the DUT and only **wire** types can be connected to the output of the module instance.

```
53 ///////////////////////////////////////////////////  
54 // DUT  
55 ///////////////////////////////////////////////////  
56 parameter WIDTH = 5;  
57 reg          start_p;  
58 reg [WIDTH -1:0] a;  
59 reg [WIDTH -1:0] b;  
60 wire          busy;  
61 wire [WIDTH*2-1:0] o;  
62  
63 mult2#(  
64     .WIDTH(WIDTH)  
65 ) I_MULT2(  
66     .start_p(start_p ),  
67     .a      (a      ),  
68     .b      (b      ),  
69     .busy   (busy   ),  
70     .o      (o      ),  
71     .clk    (clk    ),  
72     .rst_n  (rst_n  )  
73 );
```

# Drive DUT Input ■

- Control the input signals to generate test patterns to DUT
- The testbench acts as another module to interact with the DUT
- In order to describe the correct circuit interaction behavior
  - it is recommended to use edge trigger and non-blocking assignment to control these inputs.

```
82  //////////////////////////////////
83  //drive test data
84  initial begin
85
86      //////////////////////////////////
87      //reset input signal
88      @(negedge rst_n);
89      start_p <= 1'b0;
90      a      <= {WIDTH{1'b0}};
91      b      <= {WIDTH{1'b0}};
92      wait(rst_n);
93
94      //////////////////////////////////
95      //start input signal
96      repeat(3)begin
97          @(posedge clk);
98          @(posedge clk);
99          start_p <= 1'b1;
100         a      <= $random();
101         b      <= $random();
102
103         @(posedge clk);
104         start_p <= 1'b0;
```

# Monitor DUT Input and Output

- An independent monitor block to capture and store the input and output of the DUT.
- Same as driver block, you can use edge-trigger control to capture the correct data.

```
79     reg [WIDTH*2-1:0] gld_o[$]; //golden answer of o; //[$] is systemverilog queue
80     reg [WIDTH*2-1:0] dut_o[$]; //dut output      of o; //[$] is systemverilog queue
```

```
124 ///////////////////////////////////////////////////
125 //monitor data
126 initial begin
127     @(negedge rst_n);
128     wait(rst_n);
129     while(1)begin
130         @(posedge clk);
131         if(start_p)begin
132             $display($realtime,, "input  a: 'd%d",a);
133             $display($realtime,, "input  b: 'd%d",b);
134             gld_o.push_back(a*b);
135             $display($realtime,, "gld_o   : 'd%d",gld_o[$]);
136         end
137     end
138 end
```

```
139 initial begin
140     @(negedge rst_n);
141     wait(rst_n);
142     while(1)begin
143         @(posedge clk);
144         if(start_p)begin
145             while(1)begin
146                 @(posedge clk);
147                 if(!busy)begin
148                     $display($realtime,, "dut_o   : 'd%d",o);
149                     dut_o.push_back(o);
150                     break;
151                 end
152             end
153         end
154     end
155 end
```



# Check with Golden ■

- An independent checker block to compare the golden output with DUT output.
- If compare error, terminate the simulation.

```
157  //////////////////////////////////
158  //check data
159  initial begin
160      while(1)begin
161          wait(gld_o.size()>=1 && dut_o.size()>=1);
162          if(gld_o[0] === dut_o[0])begin
163              $display("check pass!!");
164              gld_o.pop_front();
165              dut_o.pop_front();
166          end
167          else begin
168              @(posedge clk);
169              @(posedge clk);
170              $display("*****");
171              $display("simulation fail!!!");
172              $display("*****");
173              $finish;
174          end
175      end
176  end
```

# Reference ■

- IEEE Std 1800™-2017
- <https://en.wikipedia.org/wiki/Verilog>
- <https://blog.csdn.net/l471094842/article/details/109714550>

Feedback to us ■

NTHU 晶片安全設計 回饋單



<https://forms.office.com/r/DYDu8vLaWN>

# Thank you!



Visit our website: [pufacademy.com](https://pufacademy.com)

Contact us: [PUFacademy@pufsecurity.com](mailto:PUFacademy@pufsecurity.com)

